where a is the increment in p. Since p=0, the first term on the left-hand side of Eq.[78] is zero. So the general formula for the reduced derivatives of s(p) is

$$S_p(n) = W_p(n+1)/a \ . \qquad [79]$$

We will discuss this method of dropping zero terms in more detail in the next section on L'Hospital rule.

Program II-8 can be used to find all the Taylor series terms of the Sine integral, si(x). Sequential differentiation of the Sine integral is obtained from Eq.[73] in conjunction with Eq.[74] and Eq.[76].

PROGRAM II-8. Differentiation of the Sine Integral, si(x).

```
      SUBROUTINE SINTX(X0,H,SI)
      DIMENSION SI(100),S(100),W(101)
C
C     X0 = INITIAL VALUE
C     H = INCREMENT
C
      XSQ = X0*X0
      HSQ = H*H
      W(1) = 0.
      W(2) = X0
      S(1) = 1.
      SI(1) = S(1)*X0
      DO 100 I=2,100
      IA = I + 1
      IZ = I - 1
      W(IA) = - W(IZ)*XSQ/(I*IZ)
      S(I) = W(IA)/X0
  100 SI(1) = SI(1) + S(I)*X0/I
C
C  THE ABOVE PORTION OF THE PROGRAM IS THE IMPLEMENTATION OF EQ.[71].
C  ALTHOUGH THE EVEN-TERMS OF THE S-SERIES ARE ZERO, THE PROGRAM IS
C    WRITTEN IN FULL FOR THE SAKE OF CLARITY, AND UNDERSTANDING.
C
C  BEGIN THE EVALUATION OF THE DERIVATIVES OF THE SINE INTEGRAL,
C    USING EQ.[73] IN CONJUNCTION WITH EQ.[74] AND EQ.[76].
C
      W(1) = SIN(X0)
      W(2) = COS(X0)*H
      S(1) = W(1)/X0
      S(2) = (W(2) - S(1)*H)/X0
      SI(2) = S(1)*H
      DO 200 I=2,99
      IA = I + 1
      IZ = I - 1
      W(IA) = - W(IZ)*HSQ/(I*IZ)
      S(IA) = (W(IA) - S(I)*H)/X0
  200 SI(IA) = S(I)*H/I
C
C  THE SI-ARRAY CONTAINS THE FIRST ONE HUNDRED TERMS OF THE TAYLOR
C    SERIES FOR THE SINE INTEGRAL, SI(X).
C
      RETURN
      END
```

PROGRAM II-9. The Sine Integral, si(y) with y = erf(x).

```
      SUBROUTINE SINTY(X0,H,SI)
      DIMENSION SI(100),S(100),W(101),U(100),Y(100),F(101)
C
C     X0 = INITIAL VALUE
C     H = INCREMENT
      C = 2./SQRT(3.1416XXXXXXXXX)
C
C  USE EQ.[55] TO EVALUATE THE ERROR FUNCTION.  FOR THIS EVALUATION, THE
C    ONLY NON-ZERO ELEMENT IN THE V-SERIES IS V(3)
C
      V3 = - X0*X0
      F(1) = 1.
      Y(1) = F(1)*X0
      F(2) = 0.
      Y(1) = Y(1) + F(2)*X0/2.
      DO 100 I=3,100
      ZI = I - 1
      IY = I - 2
      F(I) = F(IY)*V3*2./ZI
  100 Y(1) = Y(1) + F(I)*X0/I
C
C  THE ABOVE PORTION OF THE PROGRAM IS THE IMPLEMENTATION OF EQ.[55].
C  ALTHOUGH THE EVEN-TERMS OF THE F-SERIES ARE ZERO, THE PROGRAM IS
C    WRITTEN IN FULL FOR THE SAKE OF CLARITY, AND UNDERSTANDING.
C  USE EQ.[71] WITH EQ.[75] AND EQ.[79] TO EVALUATE THE SINE INTEGRAL.
C
      W(1) = 0.
      W(2) = Y(1)
      S(1) = 1.
      SI(1) = S(1)*Y(1)
      DO 150 I=2,100
      IA = I + 1
      IZ = I - 1
      W(IA) = - W(IZ)*Y(1)*Y(1)/(I*IZ)
      S(I) = W(IA)/Y(1)
  150 SI(1) = SI(1) + S(I)*Y(1)/I
C
C  THE ABOVE PORTION OF THE PROGRAM IS THE IMPLEMENTATION OF EQ.[71].
C  ALTHOUGH THE EVEN-TERMS OF THE S-SERIES ARE ZERO, THE PROGRAM IS
C    WRITTEN IN FULL FOR THE SAKE OF CLARITY, AND UNDERSTANDING.
C  BEGIN THE EVALUATION OF THE DERIVATIVES OF THE ERROR FUNCTION,
C    USING EQ.[57] IN CONJUNCTION WITH EQ.[58] AND EQ.[60].
C  AT THE SAME TIME, EVALUATE THE DERIVATIVES OF THE SINE INTEGRAL,
C    USING EQ.[83] WITH EQ.[85] AND EQ.[87].
C
      V1 = - X0*X0
      V2 = - 2.*X0*H
      V3 = - H*H
      F(1) = EXP(V1)
      Y(2) = C*F(1)*H
      F(2) = F(1)*V2
      W(1) = SIN(Y(1))
      U(1) = COS(Y(1))
      S(1) = W(1)/Y(1)
      DO 200 I=2,100
      IA = I + 1
      IZ = I - 1
      F(IA) = (F(IZ)*V3*2. + F(I)*V2)/I
      Y(I) = C*F(IZ)*H/IZ
      W(I) = ATT(IZ,U,Y,1)
      U(I) = ATT(IZ,W,Y,1)
      S(I) = (W(I) - ATS(I,Y,S,2))/Y(1)
  200 SI(I) = ATT(IZ,S,Y,1)
C
C  THE SI-ARRAY CONTAINS THE FIRST ONE HUNDRED TERMS OF THE TAYLOR
C    SERIES FOR THE SINE INTEGRAL, SI(X).
C
      RETURN
      END
```

The Sine-Integral of Another Function

We will develop the sequential differentiation of the Sine integral of a dependent variable. Just for illustration, we will make the dependent variable an Error function! Although at first this idea may seem to be impossible, it is resolved smoothly and easily. The function to be differentiated is

$$si(y) = \int_0^y \frac{\sin(p)}{p}\, dp = \int_0^y s(p)\, dp \,, \qquad [80]$$

$$\text{where} \quad y(x) := erf(x) = \frac{2}{pi^{1/2}} \int_0^x \exp(-t^2)\, dt = C \int_0^x f(t)\, dt \,, \qquad [81]$$

where C = 2/pi**(1/2). The sequential differentiation of y(x)=erf(x) was given in Program II-6 above. We wish to develop the sequential differentiation of the Sine integral of a dependent variable, si(y). In this case, the Equations [65]-[71] remain valid with the replacement of all the X's by Y's. However, there are some changes. Equation[72] becomes

$$\frac{d}{dx} si(y) = \frac{d}{dy} si(y)*\frac{dy}{dx} = s(y)*y' \,. \qquad [82]$$

In terms of reduced derivatives, Eq.[82] is

$$SI_x(2) = S_x(1)*Y_x(2) \,, \quad SI_x(m+1) = ATT(M,S,Y,1) \,. \qquad [83]$$

Equation[83] is to replace Eq.[73]. The auxiliary function w(y):=sin(y) must be accompanied by a second auxiliary function u(y):= cos(y). Their reduced derivatives with respect to x are given by

$$W(1) = \sin(Y(1)) \,, \quad W(2) = \cos(Y(1))*Y(2) = U(1)*Y(2) \,,$$

$$\text{and} \quad U(1) = \cos(Y(1)) \,, \quad U(2) = \sin(Y(1))*Y(2) = W(1)*Y(2) \,. \qquad [84]$$

The higher-order reduced derivatives of w(x) and u(x) with respect to x are given by

$$W(M+1) = ATT(M,U,Y,1) \,, \quad \text{and} \quad U(M+1) = ATT(M,W,Y,1) \,. \qquad [85]$$

Equation[85] replaces Eq.[74]. The relation between the S-array and the W-array is also changed.

$$S(1) = W(1)/Y(1) \;, \quad \text{or} \quad S(1)*Y(1) = W(1) \;. \qquad [86]$$

Then, the higher-order reduced derivatives of s(x) with respect to x are

$$S(M) = (W(M) - ATS(M,Y,S,2))/Y(1) \;. \qquad [87]$$

Equation[87] replaces Eq.[76]. Equation[75] and Eq.[79] remain valid for the Sine integral of a function with only the change of Y(1) for a.

Program II-9 is the FORTRAN implementation of Eq.[83] for the sequential differentiation of the Sine integral si(y) with y = erf(x). As one can easily see, Program II-9 is merely a combination of parts from other earlier programs.

C. L'Hospital Rule

In the automatic differentiation of analytic functions, the presence of singular points can be very troublesome. One such singular point is where the function has an indeterminate form; another is where some parameter of the function is zero. Both of these cases are easily resolved in the automatic Taylor series analysis. It will be shown in this discussion that the L'Hospital rule can be derived from Leibnitz' rule with a generalized form.

Consider the function f(x) = u(x)/v(x), where both u(x) and v(x) are equal to zero at x=a. We can find f(x) and all its derivatives at x=a in the following manner. The (m+1)st reduced derivative with respect to x of u(x) is

$$U(m+1) = V(1)*F(m+1) + V(2)*F(m) + \sum_{i=3}^{m+1} V(i)*F(m-i+2) \;, \qquad [88]$$

where we have separated the first two terms from the sum. Since V(1) in the first term on the right-hand side is zero, we can drop this term and solve For f(m).

$$F(m) = \frac{U(m+1) - \sum_{i=3}^{m+1} U(i)*F(m-i+2)}{V(2)} . \qquad [89]$$

For m=1, Eq.[89] is just the L'Hospital rule for resolving indeterminate forms.

$$f(x) = F(1) = \frac{du/dx}{dv/dx} = \frac{U(2)}{V(2)} . \qquad [90]$$

For m>1, the reduced derivatives of f(x) are resolve using Eq.[89]. It should be noted that two things have been changed from the usual ATS-sum. First, the ATS-sum for evaluating F(m) is one order higher; it is the sum For f(m+1). Second, the first term of the ATS-sum has been dropped; the term V(1)*F(m+1).

If the indeterminate form persists at the level of the first derivatives of f(x), we can resolve it by increasing the order of the ATS-sum by two and dropping the first two terms, V(1)*F(m+2) and V(2)*F(m+1). The (m+2)nd reduced derivative of u(x) is

$$U(m+2) = V(1)F(m+2) + V(2)F(m+1) + V(3)F(m) + \sum_{i=4}^{m+2} V(i)F(m-i+3). \qquad [91]$$

Since V(1)=V(2)=0, we will drop the first two terms on the right-hand side and solve for F(m).

$$F(m) = \frac{U(m+2) - \sum_{i=4}^{m+2} V(i)*F(m-i+3)}{V(3)} . \qquad [92]$$

Thus, all that is necessary to resolve an indeterminate quotient is the dropping of zero terms from the ATS-sum. The basic ATS-function subprogram can be used to resolve indeterminate quotients by simply changing indices in the function call.

We saw an example of the application of L'Hospital rule in the development of Eq.[79], where the derivatives of sin(x)/x were evaluated at x=0. Now, we will look at another interesting example of indeterminacy.

Consider the function

$$f(x) = \frac{\exp(x) - \exp(a)}{\{2*(\cosh x + a*\sinh a - \cosh a - x*\sinh a)\}^{1/2}} . \quad [93]$$

This function arises in the analysis of space-charge in semiconductor p-n junctions. The value of f(x;93) at x=a is desired. It is evident that both the numerator and the denominator are equal to zero at x=a. We let u(x) be the numerator, and s(x) be the denominator. The first derivative of u(x) with respect to x is

$$u'(x) = \exp(x) . \quad [94]$$

The first derivative of s(x) with respect to x is

$$s'(x) = \frac{\sinh x - \sinh a}{s(x)} . \quad [95]$$

The first derivative of u(x) is well behaved, but the first derivative of s(x) results in a second indeterminacy at x=a. A second application of the L'Hospital rule is needed. This leads to

$$s'(x) = \frac{\cosh x}{s'(x)} . \quad [96]$$

Solving Eq.[96] for the derivative of s(x) with respect to x at x=a, we obtain

$$s'(a) = (\cosh a)^{1/2} . \quad [97]$$

Therefore, the value of f(x) at x=a is

$$f(a) = \frac{\exp(a)}{(\cosh a)^{1/2}} . \quad [98]$$

The evaluation of the reduced derivatives of f(x) at x=a requires two applications of L'Hospital rule for each derivative. The pen and paper differentiation of this function by symbolic manipulations would quickly exhaust the patience of most humans. This tedious task becomes straight-forward book-keeping when the ATS method is applied.

Let $v(x) := s(x)*s(x)$ so that

$$v(x) = 2*(\cosh x + a*\sinh a - \cosh a - x*\sinh a) \ . \qquad [99]$$

The reduced derivatives of $v(x)$ with respect to x at $x=a$ are

$$V(1) = v(x)\Big|_{x=a} = 0 \ , \quad V(2) = 2*h*(\sinh x - \sinh a)\Big|_{x=a} = 0 \ ,$$

$$V(3) = h^2*\cosh a \ , \qquad V(4) = h^3*(\sinh a)/3 \ ,$$

and $$V(m) = \frac{V(m-2)}{(m-1)(m-2)} \ , \qquad \text{for } m > 4. \qquad [100]$$

Then, we find $s(x)$ and its derivative with respect to x at $x=a$ to be

$$S(1)*S(1) = V(1) = 0 \ , \text{ and } \ 2*S(1)*S(2) = V(2) = 0 \ . \qquad [101]$$

These two relations are not useful, because $S(1) = 0$. Neglecting the term $S(1)$ in subsequent evaluations of the reduced derivatives of $s(x)$, we find the recursive relation to be

$$S(2)*S(2) = V(3) \ ,$$

and $$S(m) = \frac{V(m+1) - \sum_{i=3}^{m-1} S(i)*S(m-i+2)}{2*S(2)} \ , \qquad \text{for } m > 2. \qquad [102]$$

Note that it is necessary to exclude four terms from the usual ATS-sum arrive at Eq.[102]. The first two terms are excluded because they have $s(1)$. The other two terms are excluded because they involve $S(m)$, which is the unknown being evaluated.

Now, we analyze the differentiation of the numerator. The reduced derivatives of $u(x)$ with respect to x at $x=a$ are given by

$$U(1) = 0 \ , \quad U(2) = \exp(a) \ , \quad \text{and} \quad U(m) = \frac{U(m-1)}{m-1} \ . \qquad [103]$$

The function $f(x)$ and all its reduced derivatives at $x=a$ are found from the expression

$$F(1)*S(1) = U(1) = 0 \ . \qquad [104]$$

The evaluation proceeds by differentiating the functions represented by The terms in Eq.[104] and excluding the terms that involve S(1) or F(m). The first differentiation yields

$$S(1)*F(2) + S(2)*F(1) = U(2) \text{ , or } F(1) = U(2)/S(2) . \qquad [105]$$

This is L'Hospital rule. The general recursive relation is

$$F(m) = \frac{U(m+1) - \sum_{i=3}^{m+1} S(i)*F(m-i+2)}{S(2)} . \qquad [106]$$

The ATS summation in Eq.[106] has two terms excluded, one with S(1) and the other with F(m).

PROGRAM II-10. Function with an Indeterminate Form, f(x;93).

```
      SUBROUTINE INDEF(A,H,F)
C
C  THIS PROGRAM EVALUATES THE FUNCTION AND REDUCED DERIVATIVES OF
C     THE FUNCTION OF EQ.[93].
C
      DIMENSION F(100),S(101),U(101),V(102)
C     A = INITIAL VALUE
C     H = INCREMENT
      EXPA = EXP(A)
      HSQ = H*H
      U(2) = EXPA*H
      V(3) = (EXPA - 1./EXPA)*HSQ/2.
      S(2) = SQRT(V(3))*H
      F(1) = U(2)/S(2)
      V(2) = (EXPA - 1./EXPA)*H
C
C  THE ABOVE ARE INITIALIZING STATEMENTS.  NOTE THAT S(1), U(1),
C     AND V(1) ARE MISSING.  ALSO NOTE THAT SINCE V(2) IS NOT NEEDED,
C     WE ARE PLACING A VALUE IN V(2) SUCH THAT V(4) CAN BE EASILY
C     EVALUATED WITHIN THE DO 100 LOOP.
C
      DO 100 I=2,100
      IA = I + 1
      IB = I + 2
      U(IA) = U(I)*H/I
      V(IB) = V(I)*HSQ/(I*IA)
      S(IA) = (V(IB) - ATS(IB,S,S,3))/(2.*S(2))
C
C  THE ABOVE STATEMENT IS EQ.[102].
C
  100 F(I) = (U(IA) - ATS(IA,S,F,3))/S(2)
C
C  THE ABOVE STATEMENT IS EQ.[106].
C
C  THE F-ARRAY CONTAINS THE FIRST ONE HUNDRED TERMS OF THE TAYLOR
C     SERIES FOR THE FUNCTION, F(X) OF EQ.[93].
C
      RETURN
      END
```

Program II-10 is the FORTRAN program for Eq.[106]. All the reduced derivatives of f(x;93) can be evaluated at x=a for a long Taylor series.

This procedure of dropping terms from the ATS-sum can also be used in the case where some parameters of a given differential equation are zero. For example, the series expansion of the Bessel function $J_n(x)$ at x=0 is impossible without this procedure. In the x-series, Eq.[36], all the terms are zero at x=0, except X(2) = h. In the Z-series, Eq.[37], all the terms are zero at x=0, except Z(3) = h*h. With X(1)=0, and Z(1)=0, Eq.[42] becomes

$$F(1)*n^2 = 0 . \qquad [107]$$

This means that F(1)=0 for all n≠0. In the case where n=0, F(1) is the leading term of the series. Since X(1)=Z(1)=Z(2)=0 and X(2)=h, Eq.[44] is simplified to

$$F(2)*(1 - n^2) = 0 . \qquad [108]$$

this means that F(2)=0 for all n≠1. In the case where n=1, F(2) is the leading term of the series. Since X(1)=Z(1)=Z(2)=0, Z(3)=h*h, and X(2)=h, Eq.[46] becomes

$$F(i) = - F(i-2)\frac{h^2}{(i-1)^2 - n^2} . \qquad [109]$$

Equation[109] is interpreted as follows. If n > i-1, the term F(i) will be zero by virtue of the fact that F(i-2)=F(i-4)=F(i-6)=...=F(2)=F(1)=0. If n=(i-1), then F(i) is the value of the leading term of the series. Therefore, F(n+1) is the leading term of the Taylor series. Finally, if n < i-1 the term F(i) has the value given by Eq.[109]. This means that

$$F(n+2) = 0 ,$$

$$F(n+3) = - F(n+1)\frac{h^2}{(n+2)^2 - n^2} = - F(n+1)\frac{h^2}{4(n+1)} ,$$

$$F(n+4) = 0 ,$$

$$F(n+5) = - F(n+3)\frac{h^2}{(n+4)^2 - n^2} = - F(n+3)\frac{h^2}{8(n+2)} , \quad \text{etc.}$$

Therefore, in terms of the leading term of the series, the Taylor expansion for the Bessel function about the point x=0 with an increment is given by

$$J_n(h) = F(n+1)\left\{ 1 - \frac{h^2}{4*(n+1)} + \frac{h^4}{8*(n+2)*4*(n+1)} - \ldots \right\} . \qquad [110]$$

Equation[110] is valid for integer as well as non-integer values of n. In short-hand form, Eq.[110] becomes

$$J_n(h) = C \frac{h^n}{n!} \sum_{i=0}^{00} (-1)^i \frac{(h/2)^{2i}}{i!\ (n+i)!} , \qquad [111]$$

Where C is a constant of the boundary condition. For non-integer values of n, Gamma functions replace the factorial functions in Eq.[111]. This series is a well-known power series for the Bessel function of the first kind. We have derived this series using the concepts in the ATS method, but without using the ATS-function subprogram. This is because Bessel's equation is a linear differential equation. Solutions of linear differential equations are functions which do not have singularities. We will analyze the power series of functions that do have singularities in the next Chapter.

When the point of the Taylor expansion for the Bessel function is not at x=0, there are no zero terms to drop. Then, one must use the full recursive relation given in Eq.[48], which was written into program II-5. This completes the discussion on L'Hospital rule.

D. Automatic Differentiation of Inverse Functions.

Up to this point, the development of the ATS method was restricted to one-dimensional Taylor series. The functions differentiated were of one independent variable. It is necessary, in this section and others following, to analyze the automatic differentiation of multi-dimensional functions. The first problem that requires multi-dimensional analysis is the differentiation of an inverse function. Given $y = f(x)$, we wish to find the derivatives of x with respect to y. The problem is to find the Taylor expansion for $x = g(y)$, where $g(y)$ is the inverse of $f(x)$. For example, given $y = \sin(x)$, the inverse function is

$$x = \sin^{-1}(y) \; ; \quad \text{so} \quad g(y) = \sin^{-1}(y) \; . \qquad [112]$$

To begin this development, we need the operator identity relating differentiations of a function with respect to y and with respect to x.

$$\frac{d}{dy} = \frac{dx}{dy} * \frac{d}{dx} \; . \qquad [113]$$

Since there is, in this discussion, differentiation with respect to two variables, the ATS-function given in Program II-1 is inadequate. We will temporarily add subscripts to the reduced derivatives to indicate the variable with respect to which the differentiation is taken. This was done earlier in the analyses of the Error function and the Sine integral. We define

$$F_{x,y}(m+1,n+1) := \frac{d^{m+n} f}{dx^m \, dy^n} * \frac{k^n}{n!} \; . \qquad [114]$$

In Eq.[114], the order m is associated with the variable x, while the order n and increment k are associated with the variable y. Note that the factor (h**m)/m! is missing from Eq.[114], when compared to Eq.[9] in Chapter I. These derivatives with respect to x are ordinary derivatives and not reduced derivatives.

In order to accommodate two-dimensional Taylor series, we will make some small changes in the ATS-function subprogram giving us a new subprogram called the ATN-function. The "N" in ATN denotes "non-linear", because the ATN-function was developed to solve non-linear equations.

PROGRAM II-11. The ATN-Function.

```
      FUNCTION ATN(N,A,MA,B,MB,J)
C
C  N IS THE ORDER OF THE SUMMATION.
C  A AND B ARE THE INPUT SERIES.
C  MA AND MB ARE THE INDICES IN X OF THE ABOVE TWO SERIES.
C  J IS THE INDEX OF THE A-SERIES TERM TO BEGIN THE SUM.
C
      DIMENSION A(MA,30),B(MB,30)
      ATN = 0.
      IF(J.GT.N) RETURN
      NA = N + 1
      DO 100 I=J,N
      L = NA - I
  100 ATN = ATN + A(MA,I)*B(MB,L)
      RETURN
      END
```

Applying the operator identity in Eq.[113] to a function f(x), and writing the result in terms of the two-dimensional reduced derivatives defined in Eq.[114], we obtain

$$F_{x,y}(1,2)\frac{1}{k} = X_y(2)\frac{1}{k}*F_{x,y}(2,1) \ . \qquad [115]$$

For the sake of simplicity, we define

$$S_y(1):= X_y(2)\frac{1}{k}, \quad \text{and} \quad S_y(n):= X_y(n+1)\frac{n}{k} \ .$$

Then, Eq.[115] becomes

$$F_{x,y}(1,2)\frac{1}{k} = 1 = S_y(1)*F_{x,y}(2,1) \ . \qquad [116]$$

The left-hand side of Eq.[116] is equal to unity. This is best shown by writing out long-hand the left-hand side.

$$F_{x,y}(1,2)\frac{1}{k} = \frac{df(x)}{dy} = \frac{dy}{dy} = 1 \ .$$